

# **esttab/estout** L<sup>A</sup>T<sub>E</sub>X tables demo

Luiza Andrade

World Bank, Development Impact Evaluations  
lcardoso@worldbank.org

May 28, 2020

## **Contents**

|          |                               |          |
|----------|-------------------------------|----------|
| <b>1</b> | <b>Set up</b>                 | <b>2</b> |
| <b>2</b> | <b>Creating simple tables</b> | <b>3</b> |
| <b>3</b> | <b>Table with two panels</b>  | <b>8</b> |

## **List of Tables**

|   |  |    |
|---|--|----|
| 1 | Basic <b>esttab</b> table . . . . .          | 3  |
| 2 | Adding variable labels . . . . .             | 4  |
| 3 | Removing omitted levels . . . . .            | 5  |
| 4 | Removing variables from table . . . . .      | 6  |
| 5 | Customizing model titles and notes . . . . . | 7  |
| 6 | Customizing header . . . . .                 | 8  |
| 7 | Table with two panels . . . . .              | 10 |

Visit <https://github.com/bbdaniels/stata-tables> to see the latest version of this material.

# 1 Set up

We'll start by loading a built-in data set and running some regressions to add to a table. The regressions themselves are not very important.

```
.
.
. // Load some data *****
.
.       sysuse census.dta, clear
(1980 Census data by state)
.
.       xtset region
       panel variable:  region (unbalanced)
.
. // Run some regressions *****
.
.       // Regression 1: nothing interesting
.       qui reg death marriage pop
.
.       est sto reg1
.       qui estadd local region "No"
.
.       // Regression 2: a different regression
.       qui reg death popurban
.
.       est sto reg2
.       qui estadd local region "No"
.
.       // Regression 3: indicator expansion
.       qui reg divorce marriage pop
.
.       est sto reg3
.       qui estadd local region "No"
.
.       // Regression 4: categorical control
.       qui reg divorce marriage pop i.region
.
.       est sto reg4
.       qui estadd local region          "Yes"
.
.       local regressions reg1 reg2 reg3 reg4
```

## 2 Creating simple tables

These regressions will be added to the same table. The table below is the simplest one that `esttab` can create.

```
.
. esttab `regressions' using "${tex}/Raw/t1_esttab_basic.tex", ///
>      replace
(output written to C:/Users/wb501238/Documents/GitHub/stata-tables/do/tex/Raw/t1_esttab_basic.tex)
.
```

| Table 1: Basic <code>esttab</code> table |                       |                       |                      |                      |
|--|-----------------------|-----------------------|----------------------|----------------------|
|  | (1)                   | (2)                   | (3)                  | (4)                  |
|  | death                 | death                 | divorce              | divorce              |
| marriage                                 | -0.0849<br>(-1.63)    |                       | 0.260***<br>(4.36)   | 0.190**<br>(3.20)    |
| pop                                      | 0.00949***<br>(19.08) |                       | 0.00271***<br>(4.74) | 0.00342***<br>(5.98) |
| popurban                                 |                       | 0.00992***<br>(28.57) |                      |                      |
| 1.region                                 |                       |                       |                      | 0<br>(.)             |
| 2.region                                 |                       |                       |                      | 6362.7*<br>(2.06)    |
| 3.region                                 |                       |                       |                      | 9215.6**<br>(3.01)   |
| 4.region                                 |                       |                       |                      | 9627.3**<br>(3.03)   |
| _cons                                    | 650.9<br>(0.48)       | 6464.6***<br>(3.55)   | -963.3<br>(-0.62)    | -7786.2**<br>(-3.00) |
| <i>N</i>                                 | 50                    | 50                    | 50                   | 50                   |

*t* statistics in parentheses

\*  $p < 0.05$ , \*\*  $p < 0.01$ , \*\*\*  $p < 0.001$

The previous table can certainly look better. Let's start by adding variable labels (using the `label` option), and replacing t-statistics with standard errors (using the `se` option).

```
.
. esttab `regressions' using "${tex}/Raw/t2_esttab_label.tex", ///
>     label ///
>     se     ///
>     replace
(output written to C:/Users/wb501238/Documents/GitHub/stata-tables/do/tex/Raw/t2_esttab_label.tex)
.
```

Table 2: Adding variable labels

|                     | (1)                      | (2)                      | (3)                      | (4)                      |
|---------------------|--------------------------|--------------------------|--------------------------|--------------------------|
|                     | Number of deaths         | Number of deaths         | Number of divorces       | Number of divorces       |
| Number of marriages | -0.0849<br>(0.0520)      |                          | 0.260***<br>(0.0597)     | 0.190**<br>(0.0593)      |
| Population          | 0.00949***<br>(0.000497) |                          | 0.00271***<br>(0.000571) | 0.00342***<br>(0.000572) |
| Urban population    |                          | 0.00992***<br>(0.000347) |                          |                          |
| NE                  |                          |                          |                          | 0<br>(.)                 |
| N Cntrl             |                          |                          |                          | 6362.7*<br>(3085.9)      |
| South               |                          |                          |                          | 9215.6**<br>(3061.1)     |
| West                |                          |                          |                          | 9627.3**<br>(3180.5)     |
| Constant            | 650.9<br>(1347.4)        | 6464.6***<br>(1819.7)    | -963.3<br>(1547.9)       | -7786.2**<br>(2591.7)    |
| Observations        | 50                       | 50                       | 50                       | 50                       |

Standard errors in parentheses

\*  $p < 0.05$ , \*\*  $p < 0.01$ , \*\*\*  $p < 0.001$

Since we have a categorical variable and a constant, one of the categories is dropped. The code below removes the base level coefficient through the `nobaselevel` option `noomit` would do the same for other omitted variables. The base category is then indicated through the `refcat` option, that add the text specified inside the parentheses above the indicated coefficient `2.region`, in this case.

The next table also shows confidence intervals instead of standard errors (option `ci` does that), and adds two custom notes.

```

.
. esttab `regressions' using "${tex}/Raw/t3_esttab_omitted.tex", ///
> ci ///
> nobaselevels ///
> refcat(2.region "Omitted category: NE region", nolabel) ///
> addnotes("Add a note here." "Other custom note here.") ///
> label ///
> replace
(output written to C:/Users/wb501238/Documents/GitHub/stata-tables/do/tex/Raw/t3_esttab_omitted.tex)
.

```

Table 3: Removing omitted levels

|                             | (1)                            | (2)                            | (3)                             | (4)                             |
|-----------------------------|--------------------------------|--------------------------------|---------------------------------|---------------------------------|
|                             | Number of deaths               | Number of deaths               | Number of divorces              | Number of divorces              |
| Number of marriages         | -0.0849<br>[-0.189,0.0196]     |                                | 0.260***<br>[0.140,0.380]       | 0.190**<br>[0.0702,0.309]       |
| Population                  | 0.00949***<br>[0.00849,0.0105] |                                | 0.00271***<br>[0.00156,0.00386] | 0.00342***<br>[0.00226,0.00457] |
| Urban population            |                                | 0.00992***<br>[0.00922,0.0106] |                                 |                                 |
| Omitted category: NE region |                                |                                |                                 |                                 |
| N Cntrl                     |                                |                                |                                 | 6362.7*<br>[143.6,12581.9]      |
| South                       |                                |                                |                                 | 9215.6**<br>[3046.4,15384.8]    |
| West                        |                                |                                |                                 | 9627.3**<br>[3217.5,16037.1]    |
| Constant                    | 650.9<br>[-2059.8,3361.6]      | 6464.6***<br>[2805.9,10123.4]  | -963.3<br>[-4077.2,2150.5]      | -7786.2**<br>[-13009.4,-2562.9] |
| Observations                | 50                             | 50                             | 50                              | 50                              |

95% confidence intervals in brackets

Add a note here.

Other custom note here.

\*  $p < 0.05$ , \*\*  $p < 0.01$ , \*\*\*  $p < 0.001$

You may also prefer to not show the control coefficients in the table at all, and just add a line in the footer indicating that they were used. The example below removes all **region** variable coefficients by through the **drop()**<sup>1</sup> option. The options **scalar** adds the footnote line. Remember we added a **region** scalar to each model when the regressions results were stored. This option uses the scalar name followed by its label in the table.

<sup>1</sup>**drop()** takes coefficient names as input. Because **region** is a categorical variable, the coefficient for each level is stored as its code plus the variable name, separated by a dot: **1.region**, **2.region**, etc.

```

. esttab `regressions' using "${tex}/Raw/t4_esttab_scalar.tex", ///
> ci ///
> drop(*.region) ///
> scalars("region Region controls") ///
> addnotes("Add a note here." "Other custom note here.") ///
> label ///
> replace
(output written to C:/Users/wb501238/Documents/GitHub/stata-tables/do/tex/Raw/t4_esttab_scalar.tex)
.

```

Table 4: Removing variables from table

|                     | (1)                            | (2)                            | (3)                             | (4)                             |
|---------------------|--------------------------------|--------------------------------|---------------------------------|---------------------------------|
|                     | Number of deaths               | Number of deaths               | Number of divorces              | Number of divorces              |
| Number of marriages | -0.0849<br>[-0.189,0.0196]     |                                | 0.260***<br>[0.140,0.380]       | 0.190**<br>[0.0702,0.309]       |
| Population          | 0.00949***<br>[0.00849,0.0105] |                                | 0.00271***<br>[0.00156,0.00386] | 0.00342***<br>[0.00226,0.00457] |
| Urban population    |                                | 0.00992***<br>[0.00922,0.0106] |                                 |                                 |
| Constant            | 650.9<br>[-2059.8,3361.6]      | 6464.6***<br>[2805.9,10123.4]  | -963.3<br>[-4077.2,2150.5]      | -7786.2**<br>[-13009.4,-2562.9] |
| Observations        | 50                             | 50                             | 50                              | 50                              |
| Region controls     | No                             | No                             | No                              | Yes                             |

95% confidence intervals in brackets

Add a note here.

Other custom note here.

\*  $p < 0.05$ , \*\*  $p < 0.01$ , \*\*\*  $p < 0.001$

Now let's add model titles. The next two tables do that. The first one adds individual titles to each model using `mtitles`, while the second groups columns through `mggroups`.

`mggroups` first two inputs are the group names. `pattern` defines which columns are part of which group (1 marks the beginning of a new group). The next line centralize group titles to include both groups, and the last line inside `mggroups` adds a line under group names.

```

.
. esttab `regressions' using "${tex}/Raw/t5_esttab_titles.tex", ///
> mtitles("Title 1" "Title 2" "Title 3" "Title 4") ///
> se ///
> drop(*.region*) ///
> scalars("region Region controls") ///
> addnotes("Add a note here." "Other custom note here.") ///
> label ///
> replace
(output written to C:/Users/wb501238/Documents/GitHub/stata-tables/do/tex/Raw/t5_esttab_titles.tex)

```

Table 5: Customizing model titles and notes

|                     | (1)                      | (2)                      | (3)                      | (4)                      |
|---------------------|--------------------------|--------------------------|--------------------------|--------------------------|
|                     | Title 1                  | Title 2                  | Title 3                  | Title 4                  |
| Number of marriages | -0.0849<br>(0.0520)      |                          | 0.260***<br>(0.0597)     | 0.190**<br>(0.0593)      |
| Population          | 0.00949***<br>(0.000497) |                          | 0.00271***<br>(0.000571) | 0.00342***<br>(0.000572) |
| Urban population    |                          | 0.00992***<br>(0.000347) |                          |                          |
| Constant            | 650.9<br>(1347.4)        | 6464.6***<br>(1819.7)    | -963.3<br>(1547.9)       | -7786.2**<br>(2591.7)    |
| Observations        | 50                       | 50                       | 50                       | 50                       |
| Region controls     | No                       | No                       | No                       | Yes                      |

Standard errors in parentheses

Add a note here.

Other custom note here.

\*  $p < 0.05$ , \*\*  $p < 0.01$ , \*\*\*  $p < 0.001$ 

```

.
. esttab `regressions' using "${tex}/Raw/t6_esttab_header.tex", ///
>     nomtitles ///
>     drop(*.region*) ///
>     scalars("region Region fixed effects") ///
>     se ///
>     label ///
>     mgroups("Number of deaths" "Number of divorces", ///
>             pattern(1 0 1 0) ///
>             span prefix{\multicolumn{@span}{c}{}} suffix{}} ///
>             erepeat{\cmidrule(lr){@span}}) ///
>     replace
(output written to C:/Users/wb501238/Documents/GitHub/stata-tables/do/tex/Raw/t6_esttab_header.tex)
.

```

Table 6: Customizing header

|                      | Number of deaths         |                          | Number of divorces       |                          |
|----------------------|--------------------------|--------------------------|--------------------------|--------------------------|
|                      | (1)                      | (2)                      | (3)                      | (4)                      |
| Number of marriages  | -0.0849<br>(0.0520)      |                          | 0.260***<br>(0.0597)     | 0.190**<br>(0.0593)      |
| Population           | 0.00949***<br>(0.000497) |                          | 0.00271***<br>(0.000571) | 0.00342***<br>(0.000572) |
| Urban population     |                          | 0.00992***<br>(0.000347) |                          |                          |
| Constant             | 650.9<br>(1347.4)        | 6464.6***<br>(1819.7)    | -963.3<br>(1547.9)       | -7786.2**<br>(2591.7)    |
| Observations         | 50                       | 50                       | 50                       | 50                       |
| Region fixed effects | No                       | No                       | No                       | Yes                      |

Standard errors in parentheses

\*  $p < 0.05$ , \*\*  $p < 0.01$ , \*\*\*  $p < 0.001$ 

### 3 Table with two panels

The `estout` and `esttab` packages can also be used to make more complex tables. However, this requires hardcoding some LaTeX into your Stata do-file.

The example below creates a table with two panels. `fragment` is used to avoid adding the whole tabular environment into LaTeX. This means that it must be added manually. That is what `prehead()` on the top panel and `postfoot()` in the bottom panel do.

The advantage of using the `fragment` option is that it allows you to append multiple tables, therefore creating the different panels. To do this, all panels but the top one must use `append` instead of `replace`. Finally, `posthead()` lets you add your panel titles.

```
.
. // South region only
. qui reg death marriage          if region == 3
. est sto s1

.
. qui reg death marriage pop      if region == 3
. est sto s2
```



```

.
. // West region only
. qui reg death marriage          if region == 4
.
. est sto w1
.
.
. qui reg death marriage pop      if region == 4
.
. est sto w2
.
.
. * Top panel
. esttab s1 s2 using "${tex}/Raw/t7_esttab_panel.tex", ///
>     prehead("\begin{tabular}{l*{2}{c}} \hline\hline") ///
>     posthead("\hline \multicolumn{3}{c}{\textbf{Panel A: South}} \\\\[1ex]") ///
>     fragment ///
>     mgroups("Number of deaths", ///
>             pattern(1 0) ///
>             span prefix(\multicolumn{@span}{c}{}) suffix({})) ///
>     nomtitles ///
>     label ///
>     replace
(output written to C:/Users/wb501238/Documents/GitHub/stata-tables/do/tex/Raw/t7_esttab_panel.tex)
.
.
. * Bottom panel
. esttab w1 w2 using "${tex}/Raw/t7_esttab_panel.tex", ///
>     posthead("\hline \multicolumn{3}{c}{\textbf{Panel B: West}} \\\\[1ex]") ///
>     fragment ///
>     append ///
>     nomtitles nonumbers noline ///
>     prefoot("\hline") ///
>     postfoot("\hline\hline \end{tabular}") ///
>     label
(output written to C:/Users/wb501238/Documents/GitHub/stata-tables/do/tex/Raw/t7_esttab_panel.tex)
.

```

Table 7: Table with two panels

|                       | Number of deaths   |                       |
|-----------------------|--------------------|-----------------------|
|                       | (1)                | (2)                   |
| <b>Panel A: South</b> |                    |                       |
| Number of marriages   | 0.622***<br>(9.27) | -0.203<br>(-1.12)     |
| Population            |                    | 0.0107***<br>(4.68)   |
| Constant              | 6626.2<br>(1.46)   | 1783.3<br>(0.58)      |
| Observations          | 16                 | 16                    |
| <b>Panel B: West</b>  |                    |                       |
| Number of marriages   | 0.737***<br>(5.98) | 0.00548<br>(0.35)     |
| Population            |                    | 0.00788***<br>(52.78) |
| Constant              | -5555.9<br>(-0.64) | -1297.2*<br>(-2.37)   |
| Observations          | 13                 | 13                    |